
Napisać program wyświetlający tabelę temperatur w skali Celsjusza i Farenheita, 
uporządkowaną w następujący sposób:

Scenariusz działania programu:

Problem

Należy odszukać formułę przeliczania pomiędzy obiema skalami temperaturowymi. 
W oparciu o źródła (np. http://p609.ha.pwr.wroc.pl/chemia/miary.htm) wynika, 
że:

temp. w st. Celsjusza = (5/9) * (temp. w st. Farenheita - 32)

oraz

temp. w st. Farenheita = (9/5) * temp. w st. Celsjusza + 32

Ułamek 5/9 w wersji dziesiętnej jest okresowy, o wartości 0.555…, zatem zasadne 
jest pozostawienie go w takiej postaci. Ułamek 9/5 można zastąpić ułamkiem 
dziesiętnym 1.8.

Przeliczniki

Instrukcje iteracyjne Instrukcje iteracyjne 

Instrukcja iteracyjna powoduje wielokrotne wykonanie pewnej grupy 
instrukcji.

Instrukcja While powtarza jedną instrukcję zapisana poniżej, dopóki wartość 
wyrażenia logicznego zapisanego pomiedzy While  a Do jest prawdziwa.

W

I

true

fals e

While W Do I

While W Do 
  I

niespełnione (fałszywe, o wartości False) 
— iteracja się kończy, 

spełnione (prawdziwe, o wartości True) 
— iteracja trwa.

Gdy wyrażenie logiczne (warunek) W jest:

Instrukcje iteracyjne Instrukcje iteracyjne 

Program TabelaTemperatur;
Var
 Celsjusz   : Integer;
 Fahrenheit : Real;
Begin
  WriteLn( 'Tabela temperatur' );
  Writeln( 'Celsjusz   Fahrenheit' );

  Celsjusz := 0;

  While Celsjusz <= 100 Do
    Begin
      Fahrenheit := 1.8 * Celsjusz + 32;
      Write( Celsjusz: 5 );
      WriteLn( Round( Fahrenheit ) : 10 );
      Celsjusz := Celsjusz + 10;
    End; (* While *)

  WriteLn( 'Nacisnij Enter by zakonczyc...' ); 
  ReadLn;
End. (* TabelaTemperatur *)

Instrukcje iteracyjne Instrukcje iteracyjne 

Celsjusz := 0;
While Celsjusz <= 100 Do
  Fahrenheit := 1.8*Celsjusz+32;
  Write( Celsjusz: 5 );
  WriteLn( Round(Fahrenheit):10 );
  Celsjusz := Celsjusz + 10;

Pozornie poprawna iteracja While… … w istocie wykona się tak:

 

Warunek tej instrukcji iteracyjnej nigdy 
nie stanie się fałszywy! Zatem ta iteracja 

teoretycznie nigdy się nie zakończy!

Instrukcje iteracyjne Instrukcje iteracyjne 

Celsjusz := 0;
While Celsjusz <= 100 Do
  Begin
    Fahrenheit := 1.8*Celsjusz+32;
    Write( Celsjusz: 5 );
    WriteLn( Round(Fahrenheit):10 );
    Celsjusz := Celsjusz + 10;
  End

Poprawna iteracja While… … wykona się tak:

Warunek tej instrukcji iteracyjnej będzie 
fałszywy, gdy wartość zmiennej Celsjusz 

stanie się większa od 100.

Zapewnia to instrukcja:

  Celsjusz := Celsjusz + 10;

Instrukcje iteracyjne Instrukcje iteracyjne 

While W Do 
  I

Instrukcja While-Do Instrukcja Repeat-Until

Repeat
  I
Until W

Służy do opisywanie iteracji ze 
sprawdzaniem warunku na początku.

Wykonaniem steruje wartość logiczna 
warunku W. 

Warunek W jest warunkiem 
kontynuacji, osiągnięcie wartości 
false przerywa iterację.

Instrukcja wewnętrzna I może nie 
wykonać się ani razu.

Służy do opisywanie iteracji ze 
sprawdzaniem warunku na końcu. 

Wykonaniem steruje wartość 
logiczna warunku W. 

Warunek W jest warunkiem 
zakończenia, osiągnięcie wartości 
true przerywa iterację.

Instrukcja wewnętrzna I wykona się 
przynajmniej raz

Instrukcje iteracyjne Instrukcje iteracyjne 

Var 
  I : Integer;
. . . 
Writeln( 'Licze do 10-ciu ...' );
I := 1;
While I <= 10 Do
  Begin
    Writeln( I, '...' );
    I := I + 1;
  End;
. . .

Var
  I : Integer;
. . . 
Writeln( 'Licze do 10-ciu ...' );
I := 1;
Repeat
   Writeln( I, '...' );
   I := I + 1;
Until I > 10;
. . .

W

I

true

false

W

I

true

false

Instrukcje iteracyjne Instrukcje iteracyjne 

Repeat
  Begin
   Writeln( I, '...' );
   I := I + 1;
  End;
Until I >10;

Można napisać tak: Ale zwykle piszemy tak:

Repeat
  Writeln( I, '...' );
  I := I + 1;
Until I >10;

Instrukcja Repeat-Until ma naturalnie określone granice, zatem stosowanie 
instrukcji złożonej w jej obrębie nie jest konieczne.

Niezależnie od typu iteracji, jej ciało musi zawierać instrukcję, która doprowadzi 
do tego, że iteracja się kiedyś zakończy.

Instrukcje iteracyjne Instrukcje iteracyjne 

Rowerzysta notuje dystanse przejechane w ramach każdego wypadu rowerowego. 
Po zakończeniu sezonu chce obliczyć, ile w sumie przejechał kilometrów oraz jaki 
był średni dystans wycieczki. Liczba dystansów nie jest z góry ustalona.

Problem

Scenariusz działania programu

Instrukcje iteracyjne Instrukcje iteracyjne 

Analiza

Program powinien wczytać kolejno przejechane dystanse, na bieżąco dodawać 
je do dystansu sumarycznego. 

Ponieważ nie wiadomo ile będzie dystansów, zakładamy, że wprowadzenie 
dystansu zerowego jest sygnałem końca wprowadzania danych. 

Po tym następuje wyświetlenie dystansu sumarycznego i średniego

Instrukcje iteracyjne Instrukcje iteracyjne 

Program DystanseUzywajacWhile;
Var
  Dystans, Suma : Real;
  Ile : Integer;
Begin
  Writeln( 'Obliczam sumaryczny i sredni dystans.' );
  Writeln( 'Podaj kolejne dystanse, 0 konczy wprowadzanie:' );

  Ile := 0;
  Suma := 0;

  Write( '>' );
  ReadLn( Dystans );

  While Dystans <> 0 Do
    Begin
      Ile := Ile + 1;
      Suma := Suma + Abs( Dystans );

      Write( '>' );
      ReadLn( Dystans );
    End; (* Ci g dalszy – nast pna strona >> *)ą ę

Instrukcje iteracyjne Instrukcje iteracyjne 

  If Ile > 0 Then
    Begin
      Writeln( 'Dystans sumaryczny: ', Suma : 6 : 2  );
      Writeln( '    Dystans sredni: ', Suma / Ile : 6 : 2  );
    End
  Else
    WriteLn( 'Nie mam nic do roboty' );
End.

Instrukcje iteracyjne Instrukcje iteracyjne 

Program DystanseUzywajacRepeat;
Var
  Dystans, Suma : Real;
  Ile : Integer;
Begin
  Writeln( 'Obliczam sumaryczny i sredni dystans.' );
  Writeln( 'Podaj kolejne dystanse, 0 konczy wprowadzanie:' );

  Ile := 0;
  Suma := 0;

  Repeat
    Write( '>' );
    ReadLn( Dystans );

    If Dystans <> 0 Then
      Begin
        Ile := Ile + 1;
        Suma := Suma + Abs( Dystans );
      End;
  Until Dystans = 0;

(* Ci g dalszy – nast pna strona >> *)ą ę

Instrukcje iteracyjne Instrukcje iteracyjne 

  If Ile > 0 Then
    Begin
      Writeln( 'Dystans sumaryczny: ', Suma : 6 : 2  );
      Writeln( '    Dystans sredni: ', Suma / Ile : 6 : 2  );
    End
  Else
    WriteLn( 'Nie mam nic do roboty' );
End.

Instrukcje iteracyjne Instrukcje iteracyjne 

Write( '>' );
ReadLn( Dystans );

While Dystans <> 0 Do
  Begin
    Ile := Ile + 1;

    Suma := Suma + Abs(Dystans);

    Write( '>' );
    ReadLn( Dystans );
  End;

Repeat
  Write( '>' );
  ReadLn( Dystans );

  If Dystans <> 0 Then
    Begin
      Ile := Ile + 1;
      Suma := Suma + Abs(Dystans);
    End;

Until Dystans = 0;

Instrukcja While-Do Instrukcja Repeat-Until

Instrukcje iteracyjne Instrukcje iteracyjne 

Odliczanie do 10-ciu z wykorzystaniem iteracji For

Program OdliczanieFor;
Var
 I : Integer;
Begin
  Writeln( 'Licze do 10-ciu ...' );
  For I := 1 To 10 Do
    WriteLn( I, '...' );

 ReadLn;
End.

Instrukcje iteracyjna For:

sterowane zakresem zadanym dla zmiennej sterującej,

zmienna sterująca musi wystąpić.

zmienna sterująca jest modyfikowana automatycznie, 

obejmuje swoim zasięgiem  jedną instrukcję po niej występującej.

Instrukcje iteracyjne Instrukcje iteracyjne 

Instrukcja iteracyjna For powoduje tylokrotne wykonanie instrukcji wewnętrznej, 
ile wartości znajduje się w przedziale wyznaczonym przez WartośćPoczątkową
i WartośćKońcową.

For Zmienna := WartośćPoczątkowa To WartośćKońcowa Do
  Instrukcja

For Zmienna := WartośćPoczątkowa DownTo WartośćKońcowa Do
  Instrukcja

Instrukcje iteracyjne Instrukcje iteracyjne 

Program OdliczanieForTo;
Var
 I : Integer;
Begin
  Writeln( 'Licze do 10-ciu ...' );
  For I := 1 To 10 Do
    WriteLn( I, '...' );

 ReadLn;
End.

Program OdliczanieForDownTo;
Var
 I : Integer;
Begin
  Writeln( 'Licze do 10-ciu ...' );
  For I := 10 DownTo 1 Do
    WriteLn( I, '...' );

 ReadLn;
End.

Licze do 10-ciu ...
1...
2...
3...
4...
5...
6...
7...
8...
9...
10...

Licze do 10-ciu ...
10...
9...
8...
7...
6...
5...
4...
3...
2...
1...

Instrukcje iteracyjne Instrukcje iteracyjne 

Zmienna, nazywana zmienną sterującą, jest typu porządkowego, 
WartośćPoczątkowa i WartośćKońcowa są typu zgodnego z typem zmiennej 
sterującej.

Zmienna sterująca musi być zadeklarowana w bloku bezpośrednio 
obejmującym instrukcje For.

WartośćPoczątkowa i WartośćKońcowa są wyznaczane dokładnie raz na 
początku wykonania instrukcji For.

Wartość zmiennej sterującej nie może być zmieniana przez instrukcję 
wewnętrzną, wartość zmiennej sterujące po zakończeniu iteracji nie jest 
określona.

Wartość zmiennej sterującej po zakończeniu iteracji może być inna niż się 
spodziewamy.

Instrukcja iteracyjna For obejmuje swoim zasięgiem jedną instrukcję po niej 
następującą.

Instrukcje iteracyjne Instrukcje iteracyjne 

Niedozwolone, bezcelowe lub ryzykowne jest:

modyfikowanie zmiennej sterującej wewnątrz iteracji,

modyfikowanie zakresów:  WartośćPoczątkowa i WartośćKońcowa,

wykorzystywanie wartości zmiennej sterującej po zakończeniu iteracji.

Var
  I, S, K : Integer;
. . .
S := 1;
K := 10;

For I := S To K Do
  Begin
    WriteLn( I );
    I := I + 1; ← Bł dą
    K := 5;     ← Bezcelowe
  End;

If I = 10 Then  ← Ryzykowne
 . . . 

Instrukcje iteracyjne Instrukcje iteracyjne 

Należy napisać program, proszący użytkownika o naciśnięcie dowolnego klawisza 
alfanumerycznego, wyświetlający następnie informacje o tym czy to jest litera 
duża, mała, czy jest to cyfra

Problem

Program powinien wczytać i zapamiętać pojedynczy znak alfanumeryczny. Potem 
powinno nastąpić określenie, z jakiego zbioru znaków on pochodzi.

Do reprezentowania w programie wartości znakowych służy typ Char.

Char — typ znakowy, obejmujący zbiór znaków używanych do komunikacji 
z człowiekiem (monitor, klawiatura, drukarka, tekstowe transfery sieciowe).

Zakres wartości: konkretny wykaz znaków oraz sposób ich uporządkowania zależy 
od implementacji. Jednak najpopularniejsze jest kodowanie znaków według ASCII 
(ang. American Standard Code for Information Interchange).

Analiza

Instrukcje iteracyjne Instrukcje iteracyjne 

Var
  C : Char;
. . .
C := 'a';
WriteLn( C );

C := '*';
WriteLn( C );

a
*

Literały znakowe

Symbole: 'a', 'b', 'c', …, 'z', '0', '1', …, '9', '*', '#', itp. to literały 
znakowe.

Literał oznacza daną wpisaną bezpośrednio w kod programu, której wartość 
wynika bezpośrednio z jej zapisu. Wartość ta nie ulega zmianie w trakcie 
działania programu.

Są m.in. literały liczbowe: 345, 550.67 itp., znakowe: 'a', '+', '&' itp., 
łańcuchowe (napisy): 'Srednie spalanie', 'Podaj liczbe' itp..

Instrukcje iteracyjne Instrukcje iteracyjne 

Funkcja Ord dla typu Char określa numer porządkowy znaku w zbiorze 
znaków. Dla kodu ASCII jest to po prostu kod znaku wg. tabeli kodów ASCII.

Funkcja Chr pozwala przekształcić liczbę całkowitą (wartość typu Integer)
w znak (o ile tej liczbie jakiś znak odpowiada). 

Jeżeli znamy kod pewnego znaku, możemy przy użyciu funkcji Chr otrzymać 
odpowiadający mu znak.

Ord( ’A’ )
65

Chr(  )65
’A’

65
Ord( 65Chr(  ) )

’A’
’A’

Chr( Ord( ’A’ ) )
65

Instrukcje iteracyjne Instrukcje iteracyjne 

Program KodZnaku;
Var
  C : Char;
Begin
  Write( 'Nacisnij klawisz alfanumeryczny: ' );

  ReadLn( C );

  Writeln( 'Znak : ''', C, ''' jego kod to : ', Ord( C ) );
End.

Instrukcje iteracyjne Instrukcje iteracyjne 

Program ZnakOKodzie;
Var
  Code : Integer;
Begin
  Write( 'Wpisz kod znaku ( 32 .. 127 ): ' );

  ReadLn( Code );

  Writeln( 'Kod ', Code, ' odpowiada znakowi ''', Chr(Code), '''' );
End.

 

Uwaga

Program ten może dziwnie się zachowywać dla znaków sterujących o kodach 
ASCII poniżej 32.

Znali sterujące, nie posiadające reprezentacji graficznej możliwej do uzyskania w 
typowym edytorze, można zapisywać w postaci wyrażonego dziesiętnie kodu 
poprzedzonego prefiksem #:

Write(#7); (* Wyprowadzenie znaku steruj cego BELL — sygnał d wi kowy *)ą ź ę

Program TypZnaku;
Var
  C : Char;
Begin
  Write( 'Nacisnij klawisz alfanumeryczny: ' );

  ReadLn( C );

  If ( C >= 'A' ) And ( C <= 'Z' ) Then
    WriteLn( 'To duza litera' )
  Else
    If ( C >= 'a' ) And ( C <= 'z' ) Then
      WriteLn( 'To mala litera' )
    Else
      If ( C >= '0' ) And ( C <= '9' ) Then
        WriteLn( 'To cyfra' )
      Else
        WriteLn( 'To nie jest znak alfanumeryczny!' );
End.

Nacisnij klawisz alfanumeryczny: R
To duza litera

Instrukcje iteracyjne Instrukcje iteracyjne 

. . .
If ( C >= 'A' ) And ( C <= 'Z' ) Then
  WriteLn( 'To duza litera' )
Else
  If ( C >= 'a' ) And ( C <= 'z' ) Then
    WriteLn( 'To mala litera' )
  Else
    If ( C >= '0' ) And ( C <= '9' ) Then
      WriteLn( 'To cyfra' )
    Else
      WriteLn( 'To nie jest znak alfanumeryczny!' );
. . .

Instrukcje iteracyjne Instrukcje iteracyjne 

Napisać program przeliczający temperaturę wyrażona w stopniach Farenheit’a na 
temperaturę w stopniach Celsjusza, tak, aby użytkownik mógł wielokrotnie 
dokonywać obliczeń bez opuszczania programu.

Problem

Po dokonaniu pojedynczej operacji przeliczenia, program powinien zapytać 
użytkownika, czy chce kontynuować obliczenia. 

Analiza

Instrukcje iteracyjne Instrukcje iteracyjne 

Program FahrenheitDoCelsjusz1;
Var
  Celsjusz, Fahrenheit: Real;
  Znak : Char;
Begin
  WriteLn( 'Stopnie Fahrenheita na Celsjusza' );

  Repeat
    Write( 'Temperatura w st. F: ' );
    ReadLn( Fahrenheit );
    Celsjusz := ( 5.0 / 9.0 ) * ( Fahrenheit - 32.0 );
    WriteLn( 'Temperatura w st. C: ', Round( Celsjusz ) );

    Write( 'Chcesz kontynuowac? (t/n) :' );
    ReadLn( Znak );
  Until ( Znak = 'N' ) Or ( Znak = 'n' );
End.

W tej wersji program kończy działanie po naciśnięciu 'N' lub 'n'. Ale wznowienie 
obliczeń odbywa się po… naciśnięciu dowolnego klawisza!

Mały szwindel

Program FahrenheitDoCelsjusz2;
Var
  Celsjusz, Fahrenheit: Real;
  Z : Char;
Begin
  WriteLn( 'Stopnie Fahrenheita na Celsjusza' );

  Repeat
    Write( 'Temperatura w st. F: ' );
    ReadLn( Fahrenheit );
    Celsjusz := ( 5.0 / 9.0 ) * ( Fahrenheit - 32.0 );
    WriteLn( 'Temperatura w st. C: ', Round( Celsjusz ) );

    Repeat
      Write( 'Chcesz kontynuowac? (t/n) :' );
      ReadLn( Z );
    Until ( Z = 'T' ) Or ( Z = 't' ) Or ( Z = 'N' ) Or ( Z = 'n' );

  Until ( Z = 'N' ) Or ( Z = 'n' );
End.

Program TypZnaku;
Var
  C : Char;
Begin
  Write( 'Nacisnij klawisz alfanumeryczny: ' );

  ReadLn( C );

  If ( C >= 'A' ) And ( C <= 'Z' ) Then
    WriteLn( 'To duza litera' )
  Else
    If ( C >= 'a' ) And ( C <= 'z' ) Then
      WriteLn( 'To mala litera' )
    Else
      If ( C >= '0' ) And ( C <= '9' ) Then
        WriteLn( 'To cyfra' )
      Else
        WriteLn( 'To nie jest znak alfanumeryczny!' );
End.

If C In ['a'..'z'] Then

If C In ['A'..'Z'] Then

If C In ['0'..'9'] Then

C In [ ’Z’ ]’A’ .. 
zbiór dużych liter

wartość logiczna

operator “należy do zbioru”

Zapis:

['0'..'9'] zastępuje: ['0', '1', '2', '3', '4', '5', '6', '7'. '8', '9' ]

oraz, np.:

['A'..'G'] zastępuje: ['A', 'B', 'C', 'D', 'E', 'F', 'G' ]

Repeat
  . . .
  Repeat
    Write( 'Chcesz kontynuowac? (t/n) :' );
    ReadLn( Z );
  Until ( Z = 'T' ) Or ( Z = 't' ) Or ( Z = 'N' ) Or ( Z = 'n' );

Until ( Z = 'N' ) Or ( Z = 'n' );

Repeat
  . . .
  Repeat
    Write( 'Chcesz kontynuowac? (t/n) :' );
    ReadLn( Z );
  Until Z In [ 'T', 't', 'N','n'];

Until Z In [ 'N', 'n' ];

Zamiast

Można napisać



  

Wielokrotne przeliczanie temperatur, wersja 3-ciaWielokrotne przeliczanie temperatur, wersja 3-cia
Podstawy programowaniaPodstawy programowania

Copyright © Roman Simiński 39Strona :

Instrukcje iteracyjne Instrukcje iteracyjne 

Program FahrenheitDoCelsjusz3;
Const
  Dzwonek = #7;
Var
  Celsjusz, Fahrenheit: Real;
  Z : Char;
Begin
  WriteLn( 'Stopnie Fahrenheita na Celsjusza' );

  Repeat
    Write( 'Temperatura w st. F: ' );
    ReadLn( Fahrenheit );
    Celsjusz := ( 5.0 / 9.0 ) * ( Fahrenheit - 32.0 );
    WriteLn( 'Temperatura w st. C: ', Round( Celsjusz ) );

    Repeat
      Write( 'Chcesz kontynuowac? (t/n) :' );
      ReadLn( Z );

      If Not Z In [ 'T', 't', 'N','n'] Then
        Write( Dzwonek );
      
    Until Z In [ 'T', 't', 'N','n'];

  Until Z In [ 'N', 'n' ];
End.

Dla dowolnego typu porządkowego określone są funkcje:

Ord — liczba porządkowa elementu w danym typie,

Succ — wyznacza następnika (następny element w typie),

Pred — wyznacza poprzednika (poprzedni element w typie).

Var
  I : Integer;
  C : Char;
. . .
C := 'A';
I := Ord( C );
Write( C, ' - ',I );
. . .

A - 65

Var
  C1 : Char;
  C2 : Char;
. . .
C1 := 'R';
C2 := Succ( C1 );
Write( C1, ' ',C2 );
. . .

R S

Var
  C1 : Char;
  C2 : Char;
. . .
C1 := 'S';
C2 := Pred( C1 );
Write( C1, ' ',C2 );
. . .

S R

Wyprowadzić do strumienia wyjściowego programu ciąg znaków składający się z 
dużych liter.

Problem

ABCDEFGHIJKLMNOPQRSTUVWXYZ

Możliwe rozwiązania

Var
  I : Integer;
. . .
For I := 65 To 90 Do
  Write( Chr( I ) );

Var
  I : Integer;
. . .
For I := Ord('A') To Ord('Z') Do
  Write( Chr( I ) );

Var
  C : Char;
. . .
For C := 'A' To 'Z' Do
  Write( C );

Rozwiązanie skuteczne, lecz 
wymagające znajomości kodów dużych 
liter.

Rozwiązanie skuteczne, nie 
wymagające znajomości kodów dużych 
liter.

Rozwiązanie skuteczne, nie 
wymagające znajomości kodów dużych 
liter, i zdecydowanie najbardziej 
eleganckie.

For Z := A To B Do
  I

Repeat
  I1; . . .; IN
Until W;

While W Do
  I

Begin
  I1; . . .; IN;
  While Not W Do
    Begin
      I1; . . .; IN
    End;
End

If W Then
  Repeat I Until Not W;

If A <= B Then
  Begin
    Z := A; I;
    While Z <> B Do
      Begin
        Z := Succ( Z ); I
      End
  End

Napisać program obliczający sumaryczny, roczny dochód. 

Program wczytuje dochody z kolejnych 12-stu miesięcy, następnie wyprowadza 
dochód sumaryczny.

Problem

Scenariusz działania programu

Program RocznyDochodFor;
Const
  LbMies = 12; 
Var
  NrMies        : Integer; 
  Dochod, Suma  : Real;    
Begin
  WriteLn( 'Obliczam sumaryczny, roczny dochod.' );
  WriteLn( 'Wprowadz dochody z kolejnych miesiecy:' );

  Dochod := 0;
  Suma := 0;

  For NrMies := 1 To LbMies Do
    Begin
      Write( NrMies : 2, ': ' );
      ReadLn( Dochod );
      Suma := Suma + Dochod;
    End;

  WriteLn( 'Sumaryczny dochod: ', Suma : 0 : 2 );
  WriteLn( 'Nacisnij Enter by zakonczyc program...' );
  ReadLn;
End.



  

Iteracja For w akcji ― sumaryczny dochód + minimum i maksimumIteracja For w akcji ― sumaryczny dochód + minimum i maksimum
Podstawy programowaniaPodstawy programowania

Copyright © Roman Simiński 45Strona :

Instrukcje iteracyjne Instrukcje iteracyjne 

Program RocznyDochodFor1;
Const
  LbMies = 12; 
Var
  NrMies : Integer;  
  Dochod, Suma, Maks, Min : Real;    
Begin
  . . .  
  Write( ' 1: ' ); ReadLn( Dochod );
  Suma := Dochod; Min  := Dochod; Maks := Dochod; 

  For NrMies := 2 To LbMies Do
   Begin
     Write( NrMies : 2, ': ' ); ReadLn( Dochod );
     Suma := Suma + Dochod;
     If Dochod > Maks Then
       Maks := Dochod;
     If Dochod < Min Then
       Min := Dochod;
   End;

  WriteLn( 'Dochody:' );
  WriteLn( ' Sumaryczny: ', Suma : 0 : 2 );
  WriteLn( ' Minimalny : ', Min  : 0 : 2 );
  WriteLn( ' Maksymalny: ', Maks : 0 : 2 );
End.

. . .
For NrMies := 1 To LbMies Do
  Begin
    Write( NrMies : 2, ': ' );
    ReadLn( Dochod );
    Suma := Suma + Dochod;
  End;
. . .

. . .
Write( '>' );
ReadLn( Dystans );
While Dystans <> 0 Do
  Begin
    Ile := Ile + 1;
    Suma := Suma + Abs( Dystans );
    Write( '>' );
    ReadLn( Dystans );
  End; 
. . .

Przetwarzanie ciągu liczbowego 
o nieznanej długości i określonym 

znaczniku końca

Przetwarzanie ciągu liczbowego 
o znanej długości

To już było — sumaryczny dystans 
wycieczek rowerowych. Liczba wycieczek 
wstępnie nieznana.

Iteracja wykonuje się dopóki nie został 
wczytany znacznik końca ciągu liczb.

Sumaryczny dochód roczny. Liczba 
miesięcy wstępnie znana.

Iteracja wykonuje się zadaną liczbę 
razy.


